

PRESEK

List za mlade matematike, fizike, astronome in računalnikarje

ISSN 0351-6652

Letnik 23 (1995/1996)

Številka 6

Strani 334-347

Martin Juvan:

O KOLEDARJU, PETKIH IN ŠTEVILU 13

Ključne besede: računalništvo, koledarji, Zellerjev obrazec, petek trinajsti.

Elektronska verzija: <http://www.presek.si/23/1278-Juvan.pdf>

© 1996 Društvo matematikov, fizikov in astronomov Slovenije

© 2010 DMFA - založništvo

Vse pravice pridržane. Razmnoževanje ali reproduciranje celote ali posameznih delov brez poprejšnjega dovoljenja založnika ni dovoljeno.

O KOLEDARJU, PETKIH IN ŠTEVILU 13

Petek, trinajstega. Saj veste, dan, ki ga je najbolje preležati v postelji. Pa še tam ne moremo biti povsem prepričani, da nas ne bo doletela nesreča. Se vam ne zdi, da je trinajsti v mesecu kar preveč pogosto petek? Da? Tudi mene že dolgo preganja ta občutek. Zato sem se odločil, da bom zadevo znanstveno raziskal.

Začnimo z opisom koledarja, ki velja v naših krajih. Gre za gregorijanski koledar, ki ga je leta 1582 v katoliških deželah vpeljal papež Gregor XIII. Danes ga uporabljajo v večini držav sveta. Po tem koledarju se leta delijo na običajna in prestopna. Vsako leto ima dvanajst mesecev. Meseci imajo po 30 ali 31 dni, izjema je le februar, ki ima v običajnih letih 28, v prestopnih pa 29 dni. Praviloma je leto, ki je deljivo s 4, prestopno. Izjema so le leta, ki so deljiva s 100, a ne s 400. Ta niso prestopna. Tako leta 1700, 1800 in 1900 niso bila prestopna, leto 2000 pa bo prestopno. Običajna leta imajo 365, prestopna pa 366 dni.

Vsi vemo, da je vsak sedmi dan petek. Med 400 zaporednimi leti je 97 prestopnih. Zato ima štiristoletno obdobje $400 \cdot 365 + 97 = 146097$ dni. To število pa je deljivo s 7, zato se naš koledar skupaj z dnevi, ob katerih nastopijo posamezni datumi, ponavlja s periodo 400 let.

Našo analizo lahko torej omejimo na obdobje 400 let. Prešteli bomo, kolikokrat v tem obdobju pride trinajsti v mesecu na posamezni dan v tednu. Seveda lahko to štetje opravimo s svinčnikom in papirjem, ampak z uporabo računalnika bo preprosteje, hitreje in bolj zanesljivo. Poznati moramo le Zellerjev obrazec, ki nam pove, na kateri dan v tednu pride izbrani datum. Nato za vseh $12 \cdot 400 = 4800$ trinajstih dni v mesecih izračunamo, na kateri dan v tednu nastopijo, seštejemo rezultate in že imamo odgovor.

Opišimo najprej, kako uporabljamo Zellerjev obrazec. Recimo, da nas zanima, kateri dan v tednu je izbrani datum $dd.mm.ll$, na primer 1. 3. 1996, pri čemer je dd dan, mm mesec in ll leto. Postavimo $d = dd$ ter $m = mm - 2$, če je $mm > 2$, in $m = mm + 10$ sicer. V zadnjem primeru tudi zmanjšamo ll za 1. Naj bo še $l = ll \bmod 100$ število, ki ga pomenita zadnji številki leta, in $s = ll \div 100$ število, ki ga dasta prvi dve številki leta, torej za 1 zmanjšano stoletje, v katerem leži izbrano leto. Nato izračunamo

$$obr = d + (13 \cdot m - 1) \div 5 + (5 \cdot l) \div 4 + (21 \cdot s) \div 4$$

in

$$dan = obr \bmod 7$$

ter ugotovimo, za kateri dan v tednu gre. Pri tem $dan = 0$ pomeni nedeljo, $dan = 1$ ponedeljek, ... in nazadnje $dan = 6$ soboto. Za naš poskusni datum 1. 3. 1996 imamo $d = 1$, $m = 1$, $l = 96$ in $s = 19$. Vstavimo v obrazec:

$$\begin{aligned} obr &= 1 + (13 \cdot 1 - 1) \text{ div } 5 + (5 \cdot 96) \text{ div } 4 + (21 \cdot 19) \text{ div } 4 \\ &= 1 + 2 + 120 + 99 = 222 \end{aligned}$$

in zatem še

$$dan = 5.$$

Prvega marca naj bi bil torej petek. Pogled na koledar in res, bil je petek!

Naslednji odstavek je nekoliko bolj zapleten in je namenjen tistim, ki jih moje zagotovilo o pravilnosti Zellerjevega obrazca ni povsem prepričalo. Ostali ga lahko brez škode preskočite.

Znanstveni pristop zahteva, da vse uporabljene prijeme natančno utemeljimo. Zato si oglejmo nekaj razlogov, zakaj Zellerjev obrazec vrne pravi dan v tednu. Takoj opazimo, da je pomemben le ostanek izraza obr pri deljenju s 7. Preverili smo že, da obrazec velja za 1. 3. 1996. Induktivno lahko pokažemo, da velja tudi za 1. marec poljubnega leta. Člen $(5 \cdot l) \text{ div } 4$ iz obrazca zapišimo kot $l + l \text{ div } 4$. Običajno leto ima 365 dni, to število pa ima ostanek 1 pri deljenju s 7. Za ustrezno povečanje števila obr za ena poskrbi člen l . Pri prestopnih letih dodatno enico pri-
nese $l \text{ div } 4$. Bolj zapleteno je, ko prestopimo v novo stoletje. Spet lahko zapišemo $(21 \cdot s) \text{ div } 4 = 5 \cdot s + s \text{ div } 4$. Če je $l = 99$, potem je $(5 \cdot l) \text{ div } 4 = 123$. Število 123 ima pri deljenju s 7 ostanek 4. Leto kasneje je $l = 0$, zato je $(5 \cdot l) \text{ div } 4 = 0$. Vendar pa se pri spremembi leta za ena poveča tudi s , to pa poveča obr za 5. Celotna sprememba je torej v običajnih letih ravno prava: $-4 + 5 = 1$. Če pa je ustrezno leto tudi prestopno, dobimo še dodatno enico iz $s \text{ div } 4$. Podobno preverimo, da obrazec velja za prvi dan poljubnega meseca. Ker imajo meseci po 30 ali 31 dni, moramo pri prehodu v nov mesec obr povečati za 2 oziroma 3. To storimo s členom $(13 \cdot m - 1) \text{ div } 5$. Izjema je le februar, ki ima le 28 oziroma 29 dni. Zanj poskrbimo tako, da se pri računanju pretvarjamo, da je februar zadnji mesec v letu ($m = mm - 2$ oziroma $m = mm + 10$ in sprememba ll). Tako koncu februarja sledi sprememba leta. Za spremembo obr pri spreminjanju dni v posameznem mesecu skrbimo s členom d .

Zapišimo še program, s katerim preštejemo, ob katerih dnevih tedna nastopajo trinajsti dnevi v posameznih mesecih.

```

program Petek.13;
{ Prešteje, ob katerih dnevih v tednu nastopijo 13. v mesecu. }
var
  dan,mesec,leto: integer; { pomožne spremenljivke za hranjenje datumov }
  stevilo: array[0..6] of integer; { stevilo[5] je število petkov trinajstega }

function Zeller(dd,mm,ll: integer): integer;
{ Vrne, kateri dan v tednu je datum dd.mm.ll. }
{ Pri tem je 0 nedelja, 1 ponedeljek, ..., 6 pa sobota. }
var
  m,l,s,obr: integer;
begin
  if mm>2 then m := mm-2
  else
    begin m := mm+10; ll := ll-1; end;
  l := ll mod 100; s := ll div 100;
  obr := dd+(13*m-1) div 5+(5*l) div 4+(21*s) div 4;
  Zeller := obr mod 7;
end; { Zeller }

begin
  for dan:=0 to 6 do stevilo[dan] := 0; { inicializacija števec }
  for leto:=1600 to 1999 do
    for mesec:=1 to 12 do begin
      dan := Zeller(13,mesec,leto);
      stevilo[dan] := stevilo[dan]+1;
    end;
  { Izpis rezultatov. }
  writeln(' Porazdelitev trinajstih dni v mesecu po dnevih tedna:');
  writeln(' ponedeljek',stevilo[1]:10);
  writeln(' torek      ',stevilo[2]:10);
  writeln(' sreda       ',stevilo[3]:10);
  writeln(' cetrtek     ',stevilo[4]:10);
  writeln(' petek      ',stevilo[5]:10);
  writeln(' sobota     ',stevilo[6]:10);
  writeln(' nedelja    ',stevilo[0]:10);
end.

```

Program izpiše naslednjo razpredelnico:

Porazdelitev trinajstih dni v mesecu po dnevih tedna:	
ponedeljek	685
torek	685
sreda	687
cetrtek	684
petek	688
sobota	684
nedelja	687

Občutek nas torej ni varal. Trinajsti v mesecu je res najpogostejše v petek. Res pa je tudi, da razlika do števila nastopov v ostalih dnevih tedna ni prav velika.

Z majhnimi spremembami gornjega programa lahko izračunamo še kopico zanimivih stvari. Tako sta v letu 1996 dva petka trinajstega: 13. september in 13. december. V letu 1997 bo en, v letu 1998 trije, v letu 1999 pa zopet le en tak petek. Preverimo tudi lahko, da je vsako leto vsaj en petek trinajstega, v nobenem letu pa niso več kot trije. Natančneje, med 400 zaporednimi leti je 171 let z enim, 170 let z dvema in 59 (nevarnih) let s po tremi petki trinajstega.

Martin Juvan